<commit_before>{ stdenv, fetchurl, pkgconfig, xkeyboard_config, libxml2, xorg
, glib, isocodes, gobjectIntrospection }:

let
  version = "5.3";
in
stdenv.mkDerivation rec {
  name = "libxklavier-${version}";

  src = fetchurl {
    url = "mirror://gnome/sources/libxklavier/${version}/${name}.tar.xz";
    sha256 = "016lpdv35z0qsw1cprdc2k5qzkdi5waj6qmr0a2q6ljn9g2kpv7b";
  };

  outputs = [ "out" "dev" "devdoc" ];

  # TODO: enable xmodmap support, needs xmodmap DB
  propagatedBuildInputs = with xorg; [ libX11 libXi xkeyboard_config libxml2 libICE glib libxkbfile isocodes ];

  nativeBuildInputs = [ pkgconfig ];

  buildInputs = [ gobjectIntrospection ];

  configureFlags = [
    "--with-xkb-base=${xkeyboard_config}/etc/X11/xkb"
    "--with-xkb-bin-base=${xorg.xkbcomp}/bin"
    "--disable-xmodmap-support"
  ];

  meta = with stdenv.lib; {
    description = "Library providing high-level API for X Keyboard Extension known as XKB";
    homepage = http://freedesktop.org/wiki/Software/LibXklavier;
    license = licenses.lgpl2Plus;
    platforms = platforms.linux;
  };
}
<commit_msg>libxklavier: 5.3 -> 5.4 (#43622)<commit_after>{ stdenv, fetchgit, autoreconfHook, pkgconfig, gtk-doc, xkeyboard_config, libxml2, xorg, docbook_xsl
, glib, isocodes, gobjectIntrospection }:

let
  version = "5.4";
in
stdenv.mkDerivation rec {
  name = "libxklavier-${version}";

  src = fetchgit {
    url = "git://anongit.freedesktop.org/git/libxklavier";
    rev = name;
    sha256 = "1w1x5mrgly2ldiw3q2r6y620zgd89gk7n90ja46775lhaswxzv7a";
  };

  outputs = [ "out" "dev" "devdoc" ];

  # TODO: enable xmodmap support, needs xmodmap DB
  propagatedBuildInputs = with xorg; [ libX11 libXi xkeyboard_config libxml2 libICE glib libxkbfile isocodes ];

  nativeBuildInputs = [ autoreconfHook pkgconfig gtk-doc docbook_xsl ];

  buildInputs = [ gobjectIntrospection ];

  preAutoreconf = ''
    export NOCONFIGURE=1
    gtkdocize
  '';

  configureFlags = [
    "--with-xkb-base=${xkeyboard_config}/etc/X11/xkb"
    "--with-xkb-bin-base=${xorg.xkbcomp}/bin"
    "--disable-xmodmap-support"
    "--enable-gtk-doc"
  ];

  meta = with stdenv.lib; {
    description = "Library providing high-level API for X Keyboard Extension known as XKB";
    homepage = http://freedesktop.org/wiki/Software/LibXklavier;
    license = licenses.lgpl2Plus;
    platforms = platforms.linux;
  };
}
{'fraction_non_alphanumeric': 0.1311336717428088, 'fraction_numerical': 0.030456852791878174, 'mean_word_length': 3.8762886597938144, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 4, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 4}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '5557367', 'n_tokens_mistral': 940, 'n_tokens_neox': 887, 'n_words': 166}